var gel1="empty";

var gel2="empty";

var gel3="empty";

var ini_end1="dia";

var ini_end2="color";

var ini_end3="2D";

var ini_rescale=25;

var ini_color2dmode=3;

var ini_affin=42;

var ini_margin=15;

var resnum=0;

//*********************************************************************************************************

macro "Initalize the macro / Set parameters [q]"{

dirm=getDirectory("macros");

string=File.openAsString(dirm+"Macro_2Dgel.ini");

//string=File.openAsString("d:\\new\\research\\codes\\ImageJ\\romannak\\Macro_2Dgel.ini");

flines = split(string, "\n");

nflines = lengthOf(flines); 

ini_end1=flines[0];

ini_end2=flines[1];

ini_end3=flines[2];

ini_rescale=flines[3];

ini_color2dmode=flines[4];

ini_affin=flines[5];

ini_margin=flines[6];

html = "<html><h1>Gel Marrage macro</h1>"

     +"<ol><li>Press 'q' - Init macro<li>load images, press 'w' - identify+preprocess images"

     +"<li>set vertical references, press 'e' - rotation<li>set ROIs, press 'r' - crop the areas for processing"

     +"<li>set the 2 reference points, press 't' - it makes the matching"

     +"<br><br><i>Comments, questions: steinbach@alga.cz";

Dialog.create("Set GelMarrage parameters");

Dialog.addString("End tag for transmission img", ini_end1);

Dialog.addString("End tag for color img", ini_end2);

Dialog.addString("End tag for 2D img", ini_end3);

Dialog.addMessage("(Other images will be closed.)")

Dialog.addSlider("Inital rescale: [%]", 10, 100, ini_rescale);

// items = newArray("3 points, fitting - NOT WORKING", "2 points - SELECT THIS!!", "1 fix + 2points for fitting  - NOT WORKING");

// Dialog.addRadioButtonGroup("Marker modes for color-2D matching", items, 3, 1, items[ini_color2dmode]);

// Dialog.addMessage("For dia-2D matching 2 points are needed.")

Dialog.addSlider("Y affine transformation: [%]", 10, 100, ini_affin);

Dialog.addSlider("Margins:", 1, 100, ini_margin);

Dialog.addHelp(html);

Dialog.show;

ini_end1=Dialog.getString();

ini_end2=Dialog.getString();

ini_end3=Dialog.getString();

ini_rescale = Dialog.getNumber();

// initmp=Dialog.getRadioButton();

// if(initmp==items[0])

//   ini_color2dmode=0;  

// else if(initmp==items[1])

//   ini_color2dmode=1;

// else if(initmp==items[2])

//   ini_color2dmode=2;

ini_color2dmode=1;

ini_affin = Dialog.getNumber();

ini_margin = Dialog.getNumber();

del=File.delete(dirm+"Macro_2Dgel.ini");

f=File.open(dirm+"Macro_2Dgel.ini");

print(f,ini_end1);

print(f,ini_end2);

print(f,ini_end3);

print(f,ini_rescale);

print(f,ini_color2dmode);

print(f,ini_affin);

print(f,ini_margin);

File.close(f);

resnum=0;

setTool("oval");

}

//*********************************************************************************************************

macro "Prepare images [w]" {

gel1="empty";

gel2="empty";

gel3="empty";

trash=0;

if(nImages<2)

  exit("Open more images for me");

for (i=1; i<=nImages; i++) {

  selectImage(i);

  pictitle=getTitle(); 

  if(endsWith(pictitle, ini_end1+".tif"))

    gel1=pictitle;

  else if(endsWith(pictitle, ini_end2+".tif"))

    gel2=pictitle;

  else if(endsWith(pictitle, ini_end3+".tif"))

    gel3=pictitle;

  else

    {rename("trash"); trash++;}

}

for (i=1; i<=trash; i++) {

  selectWindow("trash");

  close();

}

if(endsWith(gel1, ".tif")) {

  selectWindow(gel1);

  gel1=substring(gel1, 0, lengthOf(gel1)-4);

  rename(gel1);

}

if(endsWith(gel2, ".tif")) {

  selectWindow(gel2);

  gel2=substring(gel2, 0, lengthOf(gel2)-4);

  rename(gel2);

}

if(endsWith(gel3, ".tif")) {

  selectWindow(gel3);

  gel3=substring(gel3, 0, lengthOf(gel3)-4);

  rename(gel3);

}

if(gel1!="empty") {

  selectWindow(gel1);

  run("8-bit");

  run("Set Scale...", "distance=0 known=0 pixel=1 unit=pixel");

  getDimensions(width, height, channels, slices, frames);

  run("Scale...", "x="+(ini_rescale/100)+" y="+(ini_rescale/100)+" width="+(width*ini_rescale/100)+" height="+(height*ini_rescale/100)+" interpolation=Bilinear average create title=RS");

  close(gel1);

  selectWindow("RS");

  rename(gel1);

  run("Enhance Contrast...", "saturated=1");

}

if(gel2!="empty") {

  selectWindow(gel2);

  getDimensions(width, height, channels, slices, frames);

  run("Scale...", "x="+(ini_rescale/100)+" y="+(ini_rescale/100)+" width="+(width*ini_rescale/100)+" height="+(height*ini_rescale/100)+" interpolation=Bilinear average create title=RS");

  close(gel2);

  selectWindow("RS");

  run("RGB Color");

  rsname=getTitle();

  if(rsname=="RS (RGB)") {

    close("RS");

    selectWindow("RS (RGB)");

  }

  rename(gel2);

}

if(gel3!="empty") {

  selectWindow(gel3);

  run("8-bit");

  run("Set Scale...", "distance=0 known=0 pixel=1 unit=pixel");

  getDimensions(width, height, channels, slices, frames);

  run("Scale...", "x="+(ini_rescale/100)+" y="+(ini_rescale/100)+" width="+(width*ini_rescale/100)+" height="+(height*ini_rescale/100)+" interpolation=Bilinear average create title=RS");

  close(gel3);

  selectWindow("RS");

  rename(gel3);

  run("Enhance Contrast...", "saturated=1");

}

setTool("arrow");

beep();

}

//*********************************************************************************************************

macro "Correct rotation for images - vertical ref (U to D) [e]" {

if(gel1!="empty") {

  selectWindow(gel1);

  getLine(x1, y1, x2, y2, lineWidth);

  if(LineLen(x1,y1,x2,y2)>10) {

    angle=LineAngle(x1,y1,x2,y2)+90;

    run("Rotate... ", "angle="+angle+" grid=1 interpolation=Bilinear stack");

    makeLine(1, 1, 1, 1);

  }

}

if(gel2!="empty") {

  selectWindow(gel2);

  getLine(x1, y1, x2, y2, lineWidth);

  if(LineLen(x1,y1,x2,y2)>10) {

    angle=LineAngle(x1,y1,x2,y2)+90;

    run("Rotate... ", "angle="+angle+" grid=1 interpolation=Bilinear stack");

    makeLine(1, 1, 1, 1);

  }

}

if(gel3!="empty") {

  selectWindow(gel3);

  getLine(x1, y1, x2, y2, lineWidth);

  if(LineLen(x1,y1,x2,y2)>10) {

    angle=LineAngle(x1,y1,x2,y2)+90;

    run("Rotate... ", "angle="+angle+" grid=1 interpolation=Bilinear stack");

    makeLine(1, 1, 1, 1);

  }

}

setTool("rectangle");

}

//*********************************************************************************************************

macro "Crop ROI for images [r]" {

if(gel1!="empty") {

  selectWindow(gel1);

  xpoints=newArray(3);

  ypoints=newArray(3);

  if(selectionType()==0)

    getSelectionCoordinates(xpoints, ypoints);

  if(LineLen(xpoints[0],ypoints[0],xpoints[1],ypoints[1])>10) {

    run("Copy");

    run("Internal Clipboard");

    rename("CROP");

    close(gel1);

    selectWindow("CROP");

  }

  run("Rotate 90 Degrees Left");

  rename(gel1);

}

if(gel2!="empty") {

  selectWindow(gel2);

  xpoints=newArray(3);

  ypoints=newArray(3);

  if(selectionType()==0)

    getSelectionCoordinates(xpoints, ypoints);

  if(LineLen(xpoints[0],ypoints[0],xpoints[1],ypoints[1])>10) {

    run("Copy");

    run("Internal Clipboard");

    rename("CROP");

    close(gel2);

    selectWindow("CROP");

  }

  run("Rotate 90 Degrees Left");

  rename(gel2);

}

if(gel3!="empty") {

  selectWindow(gel3);

  xpoints=newArray(3);

  ypoints=newArray(3);

  if(selectionType()==0)

    getSelectionCoordinates(xpoints, ypoints);

  if(LineLen(xpoints[0],ypoints[0],xpoints[1],ypoints[1])>10) {

    run("Copy");

    run("Internal Clipboard");

    rename("CROP");

    close(gel3);

    selectWindow("CROP");

  }

  rename(gel3);

}

setTool("multipoint");

}

//*********************************************************************************************************

macro "Fit gels to eachother [t]" {

if(gel1!="empty") {

selectWindow(gel1);

getSelectionCoordinates(x1, y1);

getDimensions(width1, height1, channels, slices, frames);

}

if(gel2!="empty") {

selectWindow(gel2);

getSelectionCoordinates(x2, y2);

getDimensions(width2, height2, channels, slices, frames);

}

if(gel3!="empty") {

selectWindow(gel3);

getSelectionCoordinates(x3, y3);

getDimensions(width3, height3, channels, slices, frames);

}

if(gel1!="empty" && gel2!="empty" && gel3!="empty") {   // 3 images

  if(x1.length!=2)

    exit("Please mark 2 points on the dia gel");

  if(x2.length!=2)

    exit("Please mark 2 points on the color gel");

  if(x3.length!=2)

    exit("Please mark 2 points on the 2D gel");

  diff1=abs(x1[1]-x1[0]);

  diff2=abs(x2[1]-x2[0]);

  diff3=abs(x3[1]-x3[0]);

  if(diff1<=diff2 && diff1<=diff3)  //a nagyokat kicsinyitjuk a 1-re

  {

    selectWindow(gel1);

    run("Scale...", "x=1 y="+(ini_affin/100)+" width="+width1+" height="+(height1*ini_affin/100)+" interpolation=Bilinear average create title=RSgel1");

    xx1=x1[0];

    selectWindow(gel2);

    rescale=diff1/diff2;

    run("Scale...", "x="+rescale+" y="+(rescale*ini_affin/100)+" width="+(width2*rescale)+" height="+(height2*rescale*ini_affin/100)+" interpolation=Bilinear average create title=RSgel2");

    xx2=x2[0]*rescale;

    selectWindow(gel3);

    rescale=diff1/diff3;

    run("Scale...", "x="+rescale+" y="+rescale+" width="+(width3*rescale)+" height="+(height3*rescale)+" interpolation=Bilinear average create title=RSgel3");

    xx3=x3[0]*rescale;

  }

  else if(diff2<=diff1 && diff2<=diff3) //a nagyokat kicsinyitjuk a 2-re

  {

    selectWindow(gel1);

    rescale=diff2/diff1;

    run("Scale...", "x="+rescale+" y="+(rescale*ini_affin/100)+" width="+(width1*rescale)+" height="+(height1*rescale*ini_affin/100)+" interpolation=Bilinear average create title=RSgel1");

    xx1=x1[0]*rescale;

    selectWindow(gel2);

    run("Scale...", "x=1 y="+(ini_affin/100)+" width="+width2+" height="+(height2*ini_affin/100)+" interpolation=Bilinear average create title=RSgel2");

    xx2=x2[0];

    selectWindow(gel3);

    rescale=diff2/diff3;

    run("Scale...", "x="+rescale+" y="+rescale+" width="+(width3*rescale)+" height="+(height3*rescale)+" interpolation=Bilinear average create title=RSgel3");

    xx3=x3[0]*rescale;

  }

  else  //a nagyokat kicsinyitjuk a 3-ra

  {

    selectWindow(gel1);

    rescale=diff3/diff1;

    run("Scale...", "x="+rescale+" y="+(rescale*ini_affin/100)+" width="+(width1*rescale)+" height="+(height1*rescale*ini_affin/100)+" interpolation=Bilinear average create title=RSgel1");

    xx1=x1[0]*rescale;

    selectWindow(gel2);

    rescale=diff3/diff2;

    run("Scale...", "x="+rescale+" y="+(rescale*ini_affin/100)+" width="+(width2*rescale)+" height="+(height2*rescale*ini_affin/100)+" interpolation=Bilinear average create title=RSgel2");

    xx2=x2[0]*rescale;

    selectWindow(gel3);

    rename("RSgel3"); // alias Run Scale...:-)

    xx3=x3[0];

  }

  selectWindow("RSgel1");

  getDimensions(width1r, height1r, channels, slices, frames);

  selectWindow("RSgel2");

  getDimensions(width2r, height2r, channels, slices, frames);

  selectWindow("RSgel3");

  getDimensions(width3r, height3r, channels, slices, frames);

  xm=maxOf(xx1, maxOf(xx2, xx3));

  xqm=maxOf(width1r-xx1, maxOf(width2r-xx2, width3r-xx3));

  resname="Result"+resnum;

  resnum++;

  widthx=2*ini_margin+xm+xqm;

  heightx=4*ini_margin+height1r+height2r+height3r;

  newImage(resname, "RGB white", widthx, heightx, 1);

  selectWindow("RSgel1");

  run("Copy");

  close("RSgel1");

  selectWindow(resname);

  makeRectangle(ini_margin+xm-xx1, ini_margin, width1r, height1r);

  run("Paste");

  selectWindow("RSgel2");

  run("Copy");

  close("RSgel2");

  selectWindow(resname);

  makeRectangle(ini_margin+xm-xx2, 2*ini_margin+height1r, width2r, height2r);

  run("Paste");

  selectWindow("RSgel3");

  run("Copy");

  close("RSgel3");

  selectWindow(resname);

  makeRectangle(ini_margin+xm-xx3, 3*ini_margin+height1r+height2r, width3r, height3r);

  run("Paste");

} // 3 images

else if(gel1!="empty" && gel2=="empty" && gel3!="empty") {   // only 1-3

  if(x1.length!=2)

    exit("Please mark 2 points on the dia gel");

  if(x3.length!=2)

    exit("Please mark 2 points on the 2D gel");

  diff1=abs(x1[1]-x1[0]);

  diff3=abs(x3[1]-x3[0]);

  if(diff1<=diff3)  //a nagyokat kicsinyitjuk a 1-re

  {

    selectWindow(gel1);

    run("Scale...", "x=1 y="+(ini_affin/100)+" width="+width1+" height="+(height1*ini_affin/100)+" interpolation=Bilinear average create title=RSgel1");

    xx1=x1[0];

    selectWindow(gel3);

    rescale=diff1/diff3;

    run("Scale...", "x="+rescale+" y="+rescale+" width="+(width3*rescale)+" height="+(height3*rescale)+" interpolation=Bilinear average create title=RSgel3");

    xx3=x3[0]*rescale;

  }

  else  //a nagyokat kicsinyitjuk a 3-ra

  {

    selectWindow(gel1);

    rescale=diff3/diff1;

    run("Scale...", "x="+rescale+" y="+(rescale*ini_affin/100)+" width="+(width1*rescale)+" height="+(height1*rescale*ini_affin/100)+" interpolation=Bilinear average create title=RSgel1");

    xx1=x1[0]*rescale;

    selectWindow(gel3);

    rename("RSgel3"); // alias Run Scale...:-)

    xx3=x3[0];

  }

  selectWindow("RSgel1");

  getDimensions(width1r, height1r, channels, slices, frames);

  selectWindow("RSgel3");

  getDimensions(width3r, height3r, channels, slices, frames);

  xm=maxOf(xx1, xx3);

  xqm=maxOf(width1r-xx1, width3r-xx3);

  resname="Result"+resnum;

  resnum++;

  widthx=2*ini_margin+xm+xqm;

  heightx=3*ini_margin+height1r+height3r;

  newImage(resname, "RGB white", widthx, heightx, 1);

  selectWindow("RSgel1");

  run("Copy");

  close("RSgel1");

  selectWindow(resname);

  makeRectangle(ini_margin+xm-xx1, ini_margin, width1r, height1r);

  run("Paste");

  selectWindow("RSgel3");

  run("Copy");

  close("RSgel3");

  selectWindow(resname);

  makeRectangle(ini_margin+xm-xx3, 2*ini_margin+height1r, width3r, height3r);

  run("Paste");

} // 1-3 images

else if(gel1=="empty" && gel2!="empty" && gel3!="empty") {   // 2-3 images

  if(x2.length!=2)

    exit("Please mark 2 points on the color gel");

  if(x3.length!=2)

    exit("Please mark 2 points on the 2D gel");

  diff2=abs(x2[1]-x2[0]);

  diff3=abs(x3[1]-x3[0]);

  if(diff2<=diff3) //a nagyokat kicsinyitjuk a 2-re

  {

    selectWindow(gel2);

    run("Scale...", "x=1 y="+(ini_affin/100)+" width="+width2+" height="+(height2*ini_affin/100)+" interpolation=Bilinear average create title=RSgel2");

    xx2=x2[0];

    selectWindow(gel3);

    rescale=diff2/diff3;

    run("Scale...", "x="+rescale+" y="+rescale+" width="+(width3*rescale)+" height="+(height3*rescale)+" interpolation=Bilinear average create title=RSgel3");

    xx3=x3[0]*rescale;

  }

  else  //a nagyokat kicsinyitjuk a 3-ra

  {

    selectWindow(gel2);

    rescale=diff3/diff2;

    run("Scale...", "x="+rescale+" y="+(rescale*ini_affin/100)+" width="+(width2*rescale)+" height="+(height2*rescale*ini_affin/100)+" interpolation=Bilinear average create title=RSgel2");

    xx2=x2[0]*rescale;

    selectWindow(gel3);

    rename("RSgel3"); // alias Run Scale...:-)

    xx3=x3[0];

  }

  selectWindow("RSgel2");

  getDimensions(width2r, height2r, channels, slices, frames);

  selectWindow("RSgel3");

  getDimensions(width3r, height3r, channels, slices, frames);

  xm=maxOf(xx2, xx3);

  xqm=maxOf(width2r-xx2, width3r-xx3);

  resname="Result"+resnum;

  resnum++;

  widthx=2*ini_margin+xm+xqm;

  heightx=3*ini_margin+height2r+height3r;

  newImage(resname, "RGB white", widthx, heightx, 1);

  selectWindow("RSgel2");

  run("Copy");

  close("RSgel2");

  selectWindow(resname);

  makeRectangle(ini_margin+xm-xx2, ini_margin, width2r, height2r);

  run("Paste");

  selectWindow("RSgel3");

  run("Copy");

  close("RSgel3");

  selectWindow(resname);

  makeRectangle(ini_margin+xm-xx3, 2*ini_margin+height2r, width3r, height3r);

  run("Paste");

} // 2-3 images

if(gel1!="empty" && gel2!="empty" && gel3=="empty") {   // 1-2 images

  if(x1.length!=2)

    exit("Please mark 2 points on the dia gel");

  if(x2.length!=2)

    exit("Please mark 2 points on the color gel");

  diff1=abs(x1[1]-x1[0]);

  diff2=abs(x2[1]-x2[0]);

  if(diff1<=diff2)  //a nagyokat kicsinyitjuk a 1-re

  {

    selectWindow(gel1);

    run("Scale...", "x=1 y="+(ini_affin/100)+" width="+width1+" height="+(height1*ini_affin/100)+" interpolation=Bilinear average create title=RSgel1");

    xx1=x1[0];

    selectWindow(gel2);

    rescale=diff1/diff2;

    run("Scale...", "x="+rescale+" y="+(rescale*ini_affin/100)+" width="+(width2*rescale)+" height="+(height2*rescale*ini_affin/100)+" interpolation=Bilinear average create title=RSgel2");

    xx2=x2[0]*rescale;

  }

  else  //a nagyokat kicsinyitjuk a 2-re

  {

    selectWindow(gel1);

    rescale=diff2/diff1;

    run("Scale...", "x="+rescale+" y="+(rescale*ini_affin/100)+" width="+(width1*rescale)+" height="+(height1*rescale*ini_affin/100)+" interpolation=Bilinear average create title=RSgel1");

    xx1=x1[0]*rescale;

    selectWindow(gel2);

    run("Scale...", "x=1 y="+(ini_affin/100)+" width="+width2+" height="+(height2*ini_affin/100)+" interpolation=Bilinear average create title=RSgel2");

    xx2=x2[0];

  }

  selectWindow("RSgel1");

  getDimensions(width1r, height1r, channels, slices, frames);

  selectWindow("RSgel2");

  getDimensions(width2r, height2r, channels, slices, frames);

  xm=maxOf(xx1, xx2);

  xqm=maxOf(width1r-xx1, width2r-xx2);

  resname="Result"+resnum;

  resnum++;

  widthx=2*ini_margin+xm+xqm;

  heightx=3*ini_margin+height1r+height2r;

  newImage(resname, "RGB white", widthx, heightx, 1);

  selectWindow("RSgel1");

  run("Copy");

  close("RSgel1");

  selectWindow(resname);

  makeRectangle(ini_margin+xm-xx1, ini_margin, width1r, height1r);

  run("Paste");

  selectWindow("RSgel2");

  run("Copy");

  close("RSgel2");

  selectWindow(resname);

  makeRectangle(ini_margin+xm-xx2, 2*ini_margin+height1r, width2r, height2r);

  run("Paste");

} // 1-2 images

makeRectangle(1, 1, 0, 0);

}

//*********************************************************************************************************

//*********************************************************************************************************

macro "xxxxxxxxxxxx From here the general tools xxxxxxxxxxxxx" {

}

macro "Delete selection [d]" {

sts=IJ.getToolName();

setTool("rectangle");

makeRectangle(1, 1, 0, 0);

setTool(sts);

}

macro "Correct rotation - horizontal ref (L to R) [h]" {

setTool("arrow");

waitForUser("Set the reference horizontal line,\nfrom left to right,\nthen press OK")

getLine(x1, y1, x2, y2, lineWidth);

angle=LineAngle(x1,y1,x2,y2);

run("Rotate... ", "angle="+angle+" grid=1 interpolation=Bilinear stack");

makeLine(1, 1, 1, 1);

}

macro "Correct rotation - vertical ref (U to D) [v]" {

setTool("arrow");

waitForUser("Set the reference vertical line,\nfrom top to bottom,\nthen press OK")

getLine(x1, y1, x2, y2, lineWidth);

angle=LineAngle(x1,y1,x2,y2)+90;

run("Rotate... ", "angle="+angle+" grid=1 interpolation=Bilinear stack");

makeLine(1, 1, 1, 1);

}

macro "Flip horizontal [f]" {

run("Flip Horizontally");

}

macro "Close all images [c]" {

run("Close All");

}

macro "printdir [p]" {

print(getDirectory("macros"));

}

function LineLen(lax1,lay1,lax2,lay2)

{

return(sqrt((lax1-lax2)*(lax1-lax2)+(lay1-lay2)*(lay1-lay2)));

}

function LineAngle(lax1,lay1,lax2,lay2)

{

if(lax2-lax1>0)

  alfa=-atan((lay2-lay1)/(lax2-lax1));

else if(lax2-lax1<0)

  alfa=-atan((lay2-lay1)/(lax2-lax1))+3.1415;

else if(lay2-lay1>0)

  alfa=-3.1415/2;

else if(lay2-lay1<0)

  alfa=3.1415/2;

else

  alfa=-420;

alfa*=180/3.1415;

if(alfa>180)

  alfa-=360;

return alfa;

}

